
Wasocaml: compiling OCaml to WebAssembly

The limitations of JavaScript as the default language of the web led to the development of Wasm, a secure, efficient and modular lan-
guage. However, compiling garbage-collected languages toWasm presents challenges, including the need to compile or re-implement
the runtime. Some Wasm extensions such as Wasm-GC are developed by the Wasm working groups to facilitate the compilation of
garbage-collected languages. We present Wasocaml, an OCaml to Wasm-GC compiler. It is the first compiler for a real-world func-
tional programming language targeting Wasm-GC. Wasocaml confirms the adequacy of the Wasm-GC proposal for a functional
language and had an impact on the design of the proposal. Moreover, the compilation strategies developed within Wasocaml are ap-
plicable to other compilers and languages. Indeed, two compilers already used a design similar to ours. Finally, we describe how we
plan to handle the C/JavaScript FFIs and effect handlers, in order to allow developers to easily deploy programs mixing C, JavaScript
and OCaml code to the web, while maintaining good performances.

CCS Concepts: • Software and its engineering→ Compilers.

Additional Key Words and Phrases: OCaml, Wasm, compilation, value representation, memory, FFI

Léo Andrès, Pierre Chambart, and Jean-Christophe Filliâtre. 2024. Wasocaml: compiling OCaml to WebAssembly. 1, 1 (Septem-
ber 2024), 29 pages. https://doi.org/10.1145/nnnnnnn.nnnnnnn

1 Context

JavaScript is the de facto language of the web. However, it does show its limitations when it comes to performance,
security and safety. In order to remedy this, WebAssembly (Wasm for short) [3] has been developed as a secure modular
language of predictable performance. Its usage is expanding beyond the web, e.g. finding applications in the cloud
(Fastly, Cloudflare) and in the creation of portable binaries.

The first version of Wasm was meant to compile libraries in languages such as C, C++ or Rust to expose them for
consumption by a JavaScript program. It was designed with the explicit aim that future versions would catter to the
specific needs of other languages and uses. This version of Wasm can be seen as simplified C. Here is an example:

2 Léo Andrès, Pierre Chambart, and Jean-Christophe Filliâtre

(func $fact (param $x i32) (result i32)

(if (i32.eq (local.get $x) (i32.const 0))

(then (i32.const 1))

(else

(i32.mul

(local.get $x)

(call $fact

(i32.sub (local.get $x) (i32.const 1)))))))

AWasmmodule is executed inside what is called an embedder or a host. In the context of the web, it is a browser and
the host language is JavaScript. A Wasm program only manipulates scalar values (integers and floats) but not garbage
collected values coming from the host language, such as JavaScript objects. This would be useful in a browser context,
but requires some interaction with the GC of the embedder.

Various extensions have been in development for Wasm, some of them being part of the standard already. Our work
is based on some of them, that is to say the reference types [20], typed function references [21], GC [18], exception
handling [17] and tail calls [19] extensions.

References and GC. The aim of these proposals is to allow the program to manipulate references to different kinds
of values such as opaque objects, functions, and garbage-collected values. Since Wasm has to be memory safe, those
values are not exposed as raw pointers. Instead, the type system guarantees the proper manipulation of such values.

For instance, in the following example, the struct and the array can only be manipulated through some primitives
such as struct.get and array.get, there’s no way to access their memory address:

(type $f1 (func (param i32) (result i32)))

(type $t1 (array i31ref))

(type $t2 (struct

((field $a (ref $t1))

(field $b (ref $f1))

(field $c i32))))

(func (param $x (ref $t2)) (result i32)

(i31.get_u

(array.get $t1

(i32.const 0)

(struct.get $t2 $a (local.get $x)))))

However, Wasm is rather a compilation target than a programming language. It needs to be able to represent the
values of any kind of source language. It was not deemed possible to have a type system powerful enough to do that.
Instead, the decision was made to have a simple type system, but to rely on dynamic casts to fill the gaps, and guarantee
that those casts are efficient.

There is a sub-typing relation that controls which casts are allowed. Some types are predefined by Wasm: anyref,
eqref, i31ref, funcref and externref. Others are user defined: array and struct. There is a structural sub-typing
Manuscript submitted to ACM

Wasocaml: compiling OCaml to WebAssembly 3

relation between user types. Upcasts are implicit, downcasts are explicit and incorrect ones lead to runtime errors. It
is possible to dynamically test for type compatibility. The following figure depicts the sub-typing relation:

anyref

eqref

i31ref struct array

funcref externref

Contribution to the proposals. The general rule for the Wasm committee is to only include features with a demon-
strated use case. As there are currently very few compilers targeting the GC-proposal, some features were lacking
conclusive evidence of their usefulness. An example is the i31ref type that is not required by the Dart compiler (the
only one targeting the GC-proposal at the time). Wasocaml demonstrates the usefulness of i31ref. It also validates
the GC-proposal on a functional language. We presented Wasocaml to the Wasm-GC working group [2]. It helped in
convincing the working group to keep i31ref in the proposal.

1.0.1 Exceptions. Another serious limitation ofWasm1 is the absence of an exceptionmechanism.This is not a problem
for compiling either C or Rust, but C++ does require them. It is of course possible to encode exception, but this has a
significant runtime cost and limits the interactions with JavaScript (that has exceptions).

1.0.2 Tail Calls. Wasm1 does not allow tail calls to be optimized. This is a show-stopper when compiling functional
languages. In the tail-call proposal, new instructions such as return_call guarantee that a tail-call will be optimized.

2 Source Language

What we are interested in is compiling OCaml to Wasm. Before being able to compile, we have to choose where to
start from in the OCaml compiler. In the following subsection we will describe the possible choices and then we will
give a proper small-step semantics of the chosen language.

2.1 OCaml Intermediate Representation Choice

OCaml has many intermediate representations (IR for short). In the Lambda IR, closures are still implicit and the code
has not been optimized. The Bytecode representation is not well optimized. The Clambda one is too low-level for Wasm
because code pointers and values are mixed inside closures. Cmm is even more low-level: it has pointer arithmetic. The
Flambda IR is simpler than Flambda2, thus it is the one we choosed for our first prototype. We will use Flambda2 in the
future, as it provides better optimisations. The following figure depicts a simplified version of the OCaml compilation
chain:

Manuscript submitted to ACM

4 Léo Andrès, Pierre Chambart, and Jean-Christophe Filliâtre

Source

Lambda

Bytecode Flambda

Wasm

JavaScript ASM

Frontend

closure conversion

Native backend

Wasocaml

Bytegen

Wasm of ocaml

Js of ocaml

2.2 Native OCaml Value Representation

The various OCaml IRs are working on a uniform representation of values. There are two kinds of values: small scalars
and heap-allocated blocks.

At runtime, it is possible to test if a value is a scalar or a block. In the native backend, this is done through a technique
known as pointer tagging, but in the Flambda IR, the precise representation is not imposed yet.

Small scalars are used to represent booleans, unboxed integers, characters or constant constructors of a sum type.
Heap-allocated blocks are used to represent arrays, non-constant constructors of a sum type or pairs. For instance,
consider the following OCaml code:

let array1 = [| 1; 2; 3 |]

let array2 = [| 42; 99; 666 |]

let pair = (array1, array2)

let x = 42

In memory, it would be represented as follows:

Manuscript submitted to ACM

Wasocaml: compiling OCaml to WebAssembly 5

stack:

array1:
array2:
pair:

x: 42

heap:

1 2 3

42 99 666

Note that a given type can have values of both kinds. For instance the empty list is a scalar value but a list with at
least one element will be a heap-allocated block.

2.3 Flambda semantics

We give a formal semantics for a subset of Flambda. Compared to the full Flambda it has the following limitations: the
objects related primitives have been removed, functions only have one argument, exceptions have been removed but
we still have static exceptions that are described later.

2.3.1 Abstract syntax.

Value. A value 𝑣 is either an integer 𝑛 or an address 𝑎 .

𝑣 ::= 𝑛 (integer)

| 𝑎 (address)

An integer represents a value of many types in the OCaml source language such as an int, a bool, a char or the
constant constructors of a sum type. An address points to a heap-allocated chunk.

Store. A store M is a map from addresses to heap-allocated chunks. It is used to represent the state of the memory.
Heap-allocated chunks can either be a block made of a tag 𝑛 and a list of values 𝑣∗, a closure S .𝑥 , or a set of closures
S . A closure is made of a set of closures S and of a name 𝑥 which is its identifier inside the set.

M ::= 𝑎 ↦→ 𝑛, 𝑣∗ (block)

| S .𝑥 (closure)

| S (set of closure)

Set of closures. A set of closures S represents a set of mutually recursive functions.

S ::= V ,C

It is made of two maps V and C . The map V is the environment (the values captured by the closures). An environment
is a map from names to values.

V ::= 𝑥 ↦→ 𝑣

The map C is the closure map (the code of each function). It is a map from names to codes.

C ::= 𝑥 ↦→ 𝜆 𝑥 𝑥 . 𝑡

Manuscript submitted to ACM

6 Léo Andrès, Pierre Chambart, and Jean-Christophe Filliâtre

The code is a function that takes two arguments. The first argument is the real argument of the function. The second
one is the address of the closure being called. It is used to access the environment or to call other functions from the
same set of mutually recursive functions. The body 𝑡 of the function is a term (see below).

Binary operator.

𝑜𝑝 ::= 𝑒𝑞 (equality)

| 𝑎𝑑𝑑 (addition)

| 𝑠𝑢𝑏 (subtraction)

Branches.

B ::= 𝑛 ↦→ 𝑡

Term.

𝑡 ::= 𝑣 (value)

| 𝑥 (identifier)

| let 𝑥 = 𝑡 in 𝑡 (let-binding)

| 𝑥 𝑥 (application)

| 𝑥 <- 𝑥 (mutation)

| if 𝑥 then 𝑡 else 𝑡 (conditional)

| switch 𝑥 B B (switch)

| sraise 𝑥 𝑥∗ (static raise)

| scatch 𝑡 with 𝑥 𝑥∗ 𝑡 (static catch)

| while𝑡 𝑡 do 𝑡 (while loop)

| get_field 𝑛 𝑥 (block field read)

| set_field 𝑛 𝑥 𝑥 (block field write)

| make_block 𝑛 𝑥∗ (block creation)

| project_closure 𝑥 𝑥 (closure projection)

| project_var 𝑥 𝑥 (closure environment)

| move_within_closure 𝑥 𝑥 (closure movement)

| make_set_of_closures S (closures allocation)

| pop 𝑡 (pop environment)

| 𝑥 𝑜𝑝 𝑥 (binary operator)

Note that the pop instruction is an administrative instruction that is never present in the concrete syntax coming
from user code.
Manuscript submitted to ACM

Wasocaml: compiling OCaml to WebAssembly 7

2.3.2 Small-step semantics.

Configuration. Our semantics is operating on a configuration. A configuration is of the form V ∗,M , 𝑡 where V ∗ is
a stack of environments representing the call stack ; M is a store representing the global memory of the program ; 𝑡
is the current term being evaluated.

Reduction context.

𝐸 ::=f

| let 𝑥 = 𝐸 in 𝑡

| while𝑡 𝐸 do 𝑡

| scatch 𝐸 with 𝑥 𝑥∗ 𝑡

| pop 𝐸

Reduction inside a context.

V ∗,M , 𝑡1 → V
′∗,M

′
, 𝑡2

V ∗,M , 𝐸 (𝑡1) → V
′∗,M

′
, 𝐸 (𝑡2)

Head reductions. In order to reduce cluttering, V ∗ and M are removed from the configuration in the left-hand side
of the reduction rules when they are not read and not modified but also from the right-hand side when they are not
modified.

identifier
V0 (𝑥 ) = 𝑣

V0V
∗, 𝑥 → 𝑣

An identifier simply reduces to its image in the current environment.

let-binding
V0V

∗, let 𝑥 = 𝑣 in 𝑡

→ V0 [𝑥 ← 𝑣]V ∗, 𝑡

A let-binding let 𝑥 = 𝑣 in 𝑡 reduces to 𝑡 and adds a binding from 𝑥 to 𝑣 in the current environment.

application

V0 (𝑥1) = 𝑎 M (𝑎) = S .𝑥
S = _,C C (𝑥 ) = 𝜆 𝑥3 𝑥4 . 𝑡

V0 (𝑥2) = 𝑣

V0V
∗, M , 𝑥1 𝑥2

→ [𝑥3 ← 𝑣 ;𝑥4 ← 𝑎 ]V0V
∗, pop 𝑡

An application 𝑥1 𝑥2 looks for an address 𝑎 in the current environment. The image of 𝑎 in the store must be a closure
S .𝑥 where the set of closures S = _,C . Let the value of 𝑥 in the closure map C be 𝜆 𝑥3 𝑥4 . 𝑡 . The term reduces to pop 𝑡
and we create a new environment containing two bindings: 𝑥3 is mapped to the value of 𝑥2 in the current environment
and 𝑥4 is mapped to 𝑎 .

Manuscript submitted to ACM

8 Léo Andrès, Pierre Chambart, and Jean-Christophe Filliâtre

pop environment
V0V

∗, pop 𝑣 → V ∗, 𝑣

This simply throws away the current environment. This is used to model the call stack. A pop is inserted around each
function application and removed once the function is fully evaluated to a value.

mutate
V0 (𝑥1) = 𝑣1 V0 (𝑥2) = 𝑣2

V0V
∗, 𝑥1 <- 𝑥2 → V0 [𝑥1 ← 𝑣2]V ∗, 0

A mutation 𝑥1 <- 𝑥2 simply updates the value of 𝑥1 to be the value of 𝑥2 in the current environment. It reduces to 0

which corresponds to the value () : unit in OCaml.

conditional 1
V0 (𝑥 ) = 𝑛 𝑛 ≠ 0

V0V
∗, if 𝑥 then 𝑡1 else 𝑡2

→ 𝑡1

A conditional if 𝑥 then 𝑡1 else 𝑡2 reduces to 𝑡1 when the value of 𝑥 in the current environment is not 0 (i.e. when it
is true : bool in OCaml).

conditional 2
V0 (𝑥 ) = 𝑛 𝑛 = 0

V0V
∗, if 𝑥 then 𝑡1 else 𝑡2

→ 𝑡2

Similarly, it reduces to 𝑡2 when the value of 𝑥 is 0 (i.e. when it is false : bool in OCaml).

switch 1
V0 (𝑥 ) = 𝑛 B1 (𝑛) = 𝑡

V0V
∗, switch 𝑥 B1 B2 → 𝑡

When the value of 𝑥 in the current environment is an integer 𝑛 , then switch 𝑥 B1 B2 reduces to the image of 𝑛 in
B1.

switch 2
V0 (𝑥 ) = 𝑎 M (𝑎) = 𝑛, 𝑣∗ B2 (𝑛) = 𝑡

V0V
∗,M , switch 𝑥 B1 B2 → 𝑡

Similarly, when 𝑥 is an address 𝑎 and M (𝑎) is a block with tag 𝑛 , then switch 𝑥 B1 B2 reduces to the image of 𝑛 in
B2.

raise 1
𝑥𝑒𝑥𝑛 ≠ 𝑥

′
𝑒𝑥𝑛

scatch (sraise 𝑥𝑒𝑥𝑛 𝑥𝑛) with 𝑥 ′𝑒𝑥𝑛 𝑥
′𝑛 𝑡

→ sraise 𝑥𝑒𝑥𝑛 𝑥𝑛

If a raised exception does not match the exception expected by a static catch, then the catch is discarded and we re-raise
the exception.

Manuscript submitted to ACM

raise 2

𝑥𝑒𝑥𝑛 = 𝑥
′
𝑒𝑥𝑛 𝑥𝑛 = 𝑥0, . . . , 𝑥𝑛−1

𝑥
′𝑛 = 𝑥

′
0, . . . , 𝑥

′
𝑛−1 ∀𝑖 ∈ [0;𝑛[,V0 (𝑥𝑖 ) = 𝑣𝑖

V0V
∗,

scatch (sraise 𝑥𝑒𝑥𝑛 𝑥𝑛) with 𝑥 ′𝑒𝑥𝑛 𝑥
′𝑛 𝑡

→ V0 [𝑥
′
0 ← 𝑣0, . . . , 𝑥

′
𝑛−1 ← 𝑣𝑛−1]V ∗,

𝑡

On the contrary, when the two exceptions match, the identifiers specified in the catch are binded to the values of the
identifiers transported by the exception. The values are taken from the current environment. We have the guarantee
that if the transported identifiers were in the environment when we raised, then they are also in the environment
when we catch. This is because static exceptions never cross a function application. Finally, the reduced term is the
one attached to the catch.

raise 3
while𝑡′ (sraise 𝑥𝑒𝑥𝑛 𝑥∗) do 𝑡

→ sraise 𝑥𝑒𝑥𝑛 𝑥∗

A while simply re-raises static exceptions.

raise 4
let 𝑥 = (sraise 𝑥𝑒𝑥𝑛 𝑥∗) in 𝑡

→ sraise 𝑥𝑒𝑥𝑛 𝑥∗

Similarly, a let-binding also re-raises static exceptions.

while loop 1
𝑛 = 0

while𝑡′ 𝑛 do 𝑡 → 0

A loop while𝑡′ 𝑛 do 𝑡 reduces to 0 when 𝑛 = 0.

while loop 2
𝑛 ≠ 0

while𝑡′ 𝑛 do 𝑡

→ let _ = 𝑡 in while𝑡′ 𝑡′ do 𝑡

On the contrary, when 𝑛 ≠ 0, it reduces to let _ = 𝑡 in while𝑡′ 𝑡′ do 𝑡 . The term 𝑡′ corresponds to the original term
that was used as a condition.

field read

V0 (𝑥 ) = 𝑎 M (𝑎) = 𝑡𝑎𝑔, 𝑣∗

𝑣∗ = 𝑣0, . . . , 𝑣𝑚−1 𝑚 > 𝑛 ≥ 0

V0V
∗,M , get_field 𝑛 𝑥 → 𝑣𝑛

The term get_field 𝑛 𝑥 reduces to the 𝑛th value of the block stored at address 𝑎 with 𝑎 being the value of 𝑥 in the
current environment.

field write

V0 (𝑥1) = 𝑎 M (𝑎) = 𝑡𝑎𝑔, 𝑣∗ 𝑣∗ = 𝑣0, . . . , 𝑣𝑚−1
𝑚 > 𝑛 ≥ 0 V0 (𝑥2) = 𝑣

V0V
∗,

M ,

set_field 𝑛 𝑥1 𝑥2

→ M [𝑎 ← 𝑡𝑎𝑔, 𝑣0, . . . , 𝑣𝑛−1, 𝑣 , 𝑣𝑛+1, . . . , 𝑣𝑚−1],
0

If 𝑥1 is the address 𝑎 in the current environment, then the term set_field 𝑛 𝑥1 𝑥2 sets the 𝑛th value of the block
stored at address 𝑎 to the value of 𝑥2 in the current environment.

block creation
V0 (𝑥0) = 𝑣0, . . . ,V0 (𝑥𝑛−1) = 𝑣𝑛−1 𝑎 ∉ dom(M )

V0V
∗,M , make_block 𝑛𝑡𝑎𝑔 𝑥0, . . . , 𝑥𝑛−1

→ M [𝑎 ← 𝑛𝑡𝑎𝑔, 𝑣0, . . . , 𝑣𝑛−1], 0

This creates a new block in the store at a fresh address. The tag is given as a value while the others values of the block
are read from the current environment.

closure projection

V0 (𝑥2) = 𝑎1
M (𝑎1) = S M (𝑎2) = S .𝑥1

V0V
∗,M , project_closure 𝑥1 𝑥2

→ 𝑎2

This reduces to the address 𝑎2 of the closure which has the identifier 𝑥1 in the set of closures at address 𝑎1, with 𝑎1

being the value of 𝑥2 in the current environment. In short, it allows one to get a particular function from a previously
allocated set of closures.

variable projection

V0 (𝑥2) = 𝑎 M (𝑎) = S .𝑥
S = V , _ V (𝑥1) = 𝑣

V0V
∗,M , project_var 𝑥1 𝑥2 → 𝑣

This reduces to the value of 𝑥1 in the environment of the set of closures that contains the closure at address 𝑎𝑎 , with
𝑎𝑎 being the value of 𝑥2 in the current environment. In short, when inside a closure, it allows to read a variable from
the set of closures the closure belongs to.

closure movement

V0 (𝑥2) = 𝑎1
M (𝑎1) = S .𝑥 M (𝑎2) = S .𝑥1

V0V
∗,M , move_within_closure 𝑥1 𝑥2

→ 𝑎2

This reduces to the address 𝑎2 of the closure 𝑥1 in the set of closures to which the closure at address 𝑎1 belongs to. In
short, when inside a closure, it allows to get another closure in the current set of closures.

set alloc

S = _ , C 𝑥0 ∈ dom(C ), . . . , 𝑥𝑛−1 ∈ dom(C )
∀𝑎 ∈ [𝑎0, . . . , 𝑎𝑛], 𝑎 ∉ dom(M )

M , make_set_of_closures S

→ M [𝑎𝑛 ← S ;𝑎0 ← S .𝑥0; . . . ;𝑎𝑛−1 ← S .𝑥𝑛−1],
𝑎𝑛

This maps the set of closures S and all its closures to fresh addresses in the store. The term reduces to the address of
the set.

2.4 Example

Consider the following OCaml code:

let f = print_int

let rec iter f l =

match l with

| [] -> ()

| hd :: tl ->

f hd;

iter f tl

In Flambda, it would be represented as follows:

let f = (* ... *)

let iter =

let set =

make_closures

| cl_iter { f } env ->

let otherset =

make_set_of_closures

(* this is the closure map *)

| cl_iter_f { l } envtwo ->

switch l

| int 0 -> 0

| tag 0 ->

let x = get_field 0 l in

let f = project_var f envtwo in

let dummy = f x in

let tl = get_field 1 l in

envtwo tl

(* this is the environment *)

| f -> f

end

in

project_closure cl_iter_f otherset

in

project_closure cl_iter set

in

(* ... *)

3 Compiling Flambda to Wasm

Wasm code runs inside an embedder. In the case of the web it is the browser and the host language is JavaScript, which
has a GC. As OCaml also has a GC, it means having two different runtimes and two different garbage collectors. For
this reason, it is hard to avoid memory leaks when compiling OCaml to Wasm1. Indeed, cycles between the two GCs
can not be collected without adding a third GC. it is hard to avoid memory leaks when compiling OCaml to Wasm1,
because it means having two different runtimes and two different garbage collectors.

To properly interact with the embedder, we need to leverage the reference and the GC extensions. This extension
adds new types to the language, e.g. externref which is an opaque type representing a value from the embedder.
References cannot be stored in the linear memory of Wasm thus they cannot appear inside OCaml values when using
the previously described compilation scheme.

In order to use references, we require a completely different compilation strategy. We do not use the linear memory
but rely entirely on WasmGC. Our strategy is close to the native OCaml one, which we describe now.

We use the Flambda IR of the OCaml compiler as input for the Wasm generation. This is a step of the compilation
chain where most of the high-level OCaml-specific optimisations are already applied. Also in this IR, the closure
conversion pass is already performed. Most of the constructions of this IR maps quite directly to Wasm ones.

Wasocaml is a compiler for the full Flambda IR.We describe the full compilation scheme but for the sake of simplicity
we only formalize a subset of Flambda.

3.1 OCaml Value Representation in Wasm

As in native OCaml, we use a uniform representation. We cannot use integers, as they cannot be used as pointers. We
need to use a reference type. The most general one is anyref. We can be more precise and use eqref. This is the type
of all values that can be tested for physical equality. It is a super type of OCaml values. This also allows us to test for
equality without requiring an additional cast.

Small scalars. All OCaml values that are represented as integers in the native OCaml are represented as i31ref.
This includes the int and char types, but also all constant constructors of sum types.

Arrays. The OCaml array type is directly represented as a Wasm array.

Blocks. For other kinds of blocks, there are two possible choices: we can either use a struct with a field for the tag
and one field per OCaml value field, or use arrays of eqref with the tag stored at position 0.

3.1.1 Blocks as Structs. A block of size one is represented using the type $block1 while for size two it is $block2:

(type $block1 (struct

(field $tag i8)

(field $field0 eqref)))

(type $block2 (sub $block1) (struct

(field $tag i8)

(field $field0 eqref)

(field $field1 eqref)))

The type $block2 is declared as a sub-type of $block1 because in the OCaml IRs, the primitive for accessing a block
field is untyped: when accessing the 𝑛-th field of the block the only available information is that the block has size at
least 𝑛 + 1. It could be possible to propagate size metadata in the IRs but that wouldn’t be sufficient because of untyped
primitives such as Obj.field that we need to support in order to be compatible with existing code.

(func $snd (param $x eqref) (result eqref)

(struct.get $block2 1

(ref.cast $block2 (local.get $x))))

3.1.2 Blocks as Arrays. We represent blocks with an array of eqref:

(type $block (array eqref))

The tag is stored in the cell at index 0 of the array. Reading its value is implemented by getting the cell and casting
it to an integer:

(func $tag (param $x eqref) (result i32)

(i31.get_u (ref.cast i31ref

(array.get $block

(i32.const 0)

(ref.cast $block (local.get $x))))))

Thus accessing the field 𝑛 of the OCaml block amounts to accessing the field 𝑛 + 1 of the array:

(func $snd (param $x eqref) (result eqref)

(array.get $block

(i32.const 2)

(ref.cast $block (local.get $x))))

3.1.3 Block Representation Tradeoffs. The array representation is simpler but requires (implicit) bound checks at each
field access and a cast to read the tag.

On the other hand, the struct representation requires a more complex cast for the Wasm runtime (a sub-typing test).
A compiler propagating more types could use finer Wasm type information, providing a precise type for each struct
field. This would allow fewer casts.

OCaml blocks can be arbitrarily large, and very large ones do occur in practice. In particular, modules are compiled
as blocks and tend to be on the larger side. We have seen in the wild some examples containing thousands of fields.
Wasm only allows a sub-typing chain of length 64, which is far from sufficient for the struct encoding.

For this reason we use the blocks as arrays representation.
Manuscript submitted to ACM

3.1.4 Boxed Numbers. Raw Wasm scalars such as i64 are not sub-types of eqref thus they cannot be used directly
to represent OCaml boxed numbers. We need to box them inside a struct in order to make them compatible with our
representation:

(type $boxed_float (struct (field $v f64)))

(type $boxed_int64 (struct (field $v i64)))

3.1.5 Closures. Wasm funcref are functions, not closures, hence we need to produce values containing both the
function and its environment. The only Wasm type construction that can contain both funcref and other values are
structs. Thus, a closure is a struct containing a funcref and the captured variables. As an example, here is the type of
a closure with two captured variables:

(type $closure1 (struct

(field funcref)

(field $v1 eqref)

(field $v2 eqref)))

3.2 Control flow

Control flow has a direct equivalent with Wasm block, loop, br_table, and if instructions. Low level OCaml prim-
itives to handle exceptions are quite similar to Wasm ones. In OCaml, it is possible to generate new exceptions at
runtime by using e.g. the let exception syntax or functors and first-class modules. This is not possible in the Wasm
exception proposal. Thus, we use the same Wasm exception everywhere and manage the rest on the side by ourselves,
using an identifier to discriminate between different exceptions.

3.3 Curryfication

In OCaml, functions take only one argument. However, in pratice, functions look like they havemore than one.Without
any special management this would mean that most of the code would be functions producing closures that would
be immediately applied. To handle that, internally, OCaml does have functions taking multiple arguments, with some
special handling for timeswhen they are partially applied.This information is explicit at the Flambda level. In the native
OCaml compiler, the transformation handling that occurs in a later step called cmmgen. Hence, we have to duplicate this
in Wasocaml. Compiling this requires some kind of structural sub-typing on closures such that, closures for functions
of arity one are supertypes of all the other closures. Thankfully there are easy encodings for that in Wasm.

3.4 Stack Representation

Most of the remaining work revolves around translating from let bound expressions to a stack based language without
producing overly naive code. Also, we do not need to care too much about low level Wasm specific optimisations as
we rely on Binaryen [16] (a quite efficient Wasm to Wasm optimizer) for those.

3.5 Unboxing

The main optimisation available in native OCaml that we are missing is number unboxing. As OCaml values have a
uniform representation, only small scalars can fit in a true OCaml value. This means that the types nativeint, int32,
int64 and float have to be boxed. In numerical code, lots of intermediate values of type float are created, and in that
Manuscript submitted to ACM

case, the allocation time to box numbers completely dominates the actual computation time. To limit that problem,
there is an optimisation called unboxing performed during the cmmgen pass that tends to eliminate most of the useless
allocations. As this pass is performed after Flambda, and was not required to produce a complete working compiler,
this was left for future work. Note that the end plan is to use the next version of Flambda, which does a much better
job at unboxing. For the time being, we can expect Binaryen to perform local unboxing in some cases.

4 Formalized compilation scheme

When compiling a program, we start with the following prelude:

(module $wasocaml_generated_modul

(type $mlfun (sub final

(func (param eqref) (param eqref) (result eqref))))

(type $data (sub final

(array (mut (ref null eq)))))

(type $block (sub final (struct

(field i32)

(field (ref null $data)))))

(type $vars (sub final (array (mut eqref))))

(rec

(type $set_of_closures (sub final (struct

(field (mut (ref null $closures)))

(field (ref null $vars)))))

(type $closures (sub final

(array (mut (ref $closure)))))

(type $closure (sub final (struct

(field (ref $mlfun))

(field (mut (ref null $set_of_closures)))))))

(global $tmp_set_of_closures

(mut (ref null $set_of_closures))

ref.null $set_of_closures)

(global $switch_v

(mut i32) i32.const 0)

(elem declare (ref $mlfun)

;; to allow forward references of functions

;; ...

)

(func $start

;; ...

drop

)

(start $start))

To compile a term we define a function T from Flambda terms to Wasm expressions. We define it case by case.
Here we do not only use the S-expression notation for Wasm but also use the stack notation which makes these rules
clearer.

T (𝑛) = i32.const 𝑛

i31.new

An integer literal is turned into an i31ref in order to match with the uniform representation.

T (𝑥 ) = local.get $𝑥

To access the value of an identifier, we simply access the value stored in the local variable of the current function.

T (let 𝑥 = 𝑡1 in 𝑡2) = T (𝑡1)

local.set $𝑥

T (𝑡2)

A let-binding is compiled by evaluating 𝑡1, storing its result in the local 𝑥 and evaluating 𝑡2.

T (𝑥1 𝑥2) = local.get $𝑥1

ref.cast (ref $closure)

local.get $𝑥2

ref.as_non_null

local.get $𝑥1

ref.cast (ref $closure)

struct.get $closure 0

call_ref $mlfun

To apply a function 𝑥1 to a value 𝑥2 we first put the two parameters of the function on the stack. The first one is the
closure itself, which is needed to access the environment and other functions in the set of closures. The second one
can be any value of type eqref. Then we need to put the function of the stack. This is done by accessing the field 0 of
the closure, which is a funcref. Then we use call_ref to call the function with its two arguments. The type of the
function reference is always $mlfun.

T (𝑥1 <- 𝑥2) = local.get $𝑥2

local.set $𝑥1

i32.const 0

i31.new

A mutation simply puts the content of 𝑥2 into 𝑥1 and leaves 0 on the stack (this is () : unit in OCaml).

T (if 𝑥1 then 𝑡1 else 𝑡2) = local.get $𝑥1

ref.cast i31ref

i31.get_s

(if (result (ref null eq))

(then T (𝑡1))

(else T (𝑡2)))

ref.as_non_null

A conditional converts the value of 𝑥1 to an i32 and then uses a Wasm if. The two branches are simply the result of
compiling 𝑡1 and 𝑡2.

T (switch 𝑥 B1 B2) =

(block $switch_a (result (ref null eq))

(block $switch_b (result (ref null eq))

local.get $𝑥

br_on_cast $switch_b (ref null eq) i31ref

;; handle block case

ref.cast (ref $block)

struct.get $block 0

global.set $switch_v

F (B2)

;; skip the scalar case

br $switch_a)

;; handle scalar case

ref.cast (ref i31)

i31.get_s

global.set $switch_v

F (B1))

To compile a switch, we start by testing if 𝑥 is a block or a scalar value. If it’s a block, we store its tag, otherwise we
store the value directly. Then, we use an auxiliary function F in order to compile the various cases of each branches.
Let 𝑖0, . . . , 𝑖𝑛−1 ∈ dom(B) be such that � 𝑗, 𝑗 ∉ dom(B) ∧ ∀𝑘 ∈ [0;𝑛[, 𝑗 ≠ 𝑖𝑘 . Then we define the auxiliary function as
follows:

F (B) =

(i32.eq (i32.const 𝑖𝑛−1) (global.get $switch_v))

(if (result (ref null eq)) (then T (B (𝑖𝑛−1)))

(else

;; ...

(i32.eq (i32.const 𝑖0) (global.get $switch_v))

(if (result (ref null eq)) (then T (B (𝑖0)))

(else unreachable))))

We have a sequence of nested if to find which of the binding in B we should evaluate. If none of them correspond to
the value stored in $switch_v (which has been set by the code generated when compiling the switch) then we have
to put an unreachable in order to generate Wasm code that typechecks. In pratice, the unreachable corresponds to
a match failure, which should not happen in a valid source program, thus it is indeed unreachable.

T (sraise 𝑥𝑒𝑥𝑛 𝑥𝑛) = local.get $𝑥𝑛−1

. . .

local.get $𝑥0

br $exn_𝑥𝑒𝑥𝑛

A static raise puts all the value transported by the exception on the stack and branches to the block corresponding to
the raise.

T (scatch 𝑡1 with 𝑥𝑒𝑥𝑛 𝑥𝑛 𝑡2) =

(block $after_try_catch (result (ref null eq))

(block $exn_𝑥𝑒𝑥𝑛

(result (ref null eq)) ;; 0

(result (ref null eq)) ;; ...

(result (ref null eq)) ;; n - 1

T (𝑡1)

br $after_try_catch)

local.set $𝑥0

. . .

local.set $𝑥𝑛−1

T (𝑡2) )

A catch has two nested blocks. The first one is used in case no exception is raised and directly returns the value of 𝑡1.
The second one corresponds to the case where an exception is raised. It needs to have many results corresponding to
all the values transported by the exception. It will put all theses values in the right locals.

T (while𝑡1 𝑡1 do 𝑡2) = (block $exit_while_loop

(loop $while_loop

T (𝑡1)

i31.get_s

i32.eqz

br_if $exit_while_loop

T (𝑡2)

drop

br $while_loop ))

i32.const 0

i31.new

A while loop maps quite directly to a Wasm loop. We also need an other block in order to exit the loop when the
condition is false.

T (get_field 𝑛 𝑥 ) = local.get $𝑥

ref.cast (ref $block)

struct.get $block 1

ref.as_non_null

i32.const 𝑛

array.get $data

To read a field of a block, we need to cast it to a block, access its second field which contains the values of the block
(the first field being the tag). Then we simply read the element at the index 𝑛 .

T (set_field 𝑛 𝑥1 𝑥2) = local.get $𝑥1

ref.cast (ref $block)

struct.get $block 1

ref.cast (ref $data)

i32.const 𝑥

local.get $𝑥2

array.set $data

i32.const 0

i31.new

Similarly, to set a field of a block, we cast it, access its data field and then set the element at index 𝑛 to the value of 𝑥2.
The array is mutable and there is no need to update the structure after the element has been set.

T (make_block 𝑛 𝑥0, . . . , 𝑥𝑚−1) = i32.const 𝑛

local.get $𝑥0

. . .

local.get $𝑥𝑚−1

array.new_fixed $data𝑚

struct.new $block

To create a new block, we put its tag on the stack. Then we put all its data and make a new array from it. With the tag
and the array we then make a new structure of type $block.

T (project_closure 𝑥1 𝑥2) =

local.get $𝑥2

ref.cast (ref $set_of_closures)

struct.get $set_of_closures 0

global.get $closure_offset_within_set_𝑥1

array.get $closures

To read a closure from a set of closures, we cast 𝑥2 to a set. Then we access its first field, which is an array of closures.
Then we look for the offset of 𝑥1 and access it.

T (project_var 𝑥1 𝑥2) =

local.get $𝑥2

ref.cast (ref $closure)

struct.get $closure 1

ref.as_non_null

struct.get $set_of_closures 1

global.get $var_offset_within_set_𝑥1

array.get $vars

This is similar to the closure case but the difference here is that 𝑥2 is a closure and not a set. We need to access its
parent first. Then we access the second field, which is the array of the captured variables. Then we need to look for the
offset of the variable again before reading the array.

T (move_within_closure 𝑥1 𝑥2) =

local.get $𝑥2

ref.cast (ref $closure)

struct.get $closure 1

ref.as_non_null

struct.get $set_of_closures 0

global.get $closure_offset_within_set_𝑥1

array.get $closures

This is the same as the previous case: 𝑥2 is a closure and not a set. The only difference being that we look in the first
field instead of the second one.

For allocating a set of closuresS = V ,C we define𝑥0, ..., 𝑥𝑛−1 to be the values that are captured inV and 𝑓0, ..., 𝑓𝑚−1
to be the names of the functions that are part of the set.

T (make_set_of_closures S ) =

;; empty closures

ref.null $closures

;; generating variables

(ref.as_non_null (local.get $𝑥0))

;; ...

(ref.as_non_null (local.get $𝑥𝑛−1))

array.new_fixed $vars 𝑛

;; set of closures with empty closures and filled vars

struct.new $set_of_closures

global.set $tmp_set_of_closures

global.get $tmp_set_of_closures

;; generating closures

ref.func $𝑓0

global.get $tmp_set_of_closures

struct.new $closure

;; ...

ref.func $𝑓𝑚−1

global.get $tmp_set_of_closures

struct.new $closure

array.new_fixed $closures𝑚

;; patching the set with closures

struct.set $set_of_closures 0

;; returning the set

global.get $tmp_set_of_closures

We start by generating an empty array of closures.We build the array made of the captured variables.Thenwe generate
an empty set of closures from our empty array of closures and our array of variables.. Each closure needs a reference
to its set, hence we have to store the set in a global before constructing them. Then, we build the array made of the
various functions in the set and patch the set.

The code of the compiled Flambda program is inserted as the code of the $start function. But we also need to
analyse the term in order to do a few more things. Each function defined in a set of closures is also added as a new
Wasm function and marked as a declarative element (in order to be referenced by the ref.func instruction). More-
over, each function and each variable in a set of closures has its own offset stored as a global variable (for instance
global $closure_offset_within_set_myfunc). Finally, each function is analysed in order to find which locals it
requires to be declared.

For now, we don’t have a proof of the correcteness of our compilation scheme. The formal semantics of the various
Wasm extensions we are needing is still an ongoing work by the various working groups. We would like to use these
semantics in the future.

4.1 Example of generated code

The code of cl_iter_f from the example given in 2.4 would be compiled to the following Wasm code:

(func $cl_iter_f (param $envtwo eqref) (param $l eqref)

(result eqref) (local $x eqref) (local $f eqref)

(local $dummy eqref) (local $tl eqref)

(block $switch_a (result (ref null eq))

(block $switch_b (result (ref null eq))

local.get $l

br_on_cast $switch_b (ref null eq) i31ref

ref.cast (ref $block)

struct.get $block 0

global.set $switch_v

global.get $switch_v

i32.const 0

i32.eq

(if (result (ref null eq))

(then

local.get $l

ref.cast (ref $block)

struct.get $block 1

;; ...

local.get $envtwo

ref.cast (ref $closure)

struct.get $closure 0

call_ref $mlfun)

(else unreachable))

br $switch_a)

ref.cast (ref i31)

i31.get_s

global.set $switch_v

global.get $switch_v

i32.const 0

i32.eq

(if (result (ref null eq))

(then

i32.const 0

i31.new)

(else

unreachable))))

5 Performances

We are able to run the classical functional microbenchmarks using the experimental branch of V8 and the results are
quite convincing.While we do not expect to be competitivewith the native code compiler, the performance degradation
is almost constant (around twice times slower).

We are also able to compile anOCaml implementation of the Knuth-Bendix algorithm [5]. OnV8 theWasm exception
handling proposal is really slow compared to native OCaml: a raise is around a hundred times slower. In SpiderMonkey
(the Firefox engine) they are much faster but other proposals are missing and we can’t run code we generate. We have
to discuss with the V8 team if this is expected and if this can be improved. We implemented an alternative compilation
strategy for exceptions where we do not useWams exceptions. Insteadwemake every function return a pair containing
its return value along with a boolean indicating if an exception was raised or not. When using this alternative strategy,
we are around two times slower than native OCaml.

The V8 runtime allows to consider casts as no-ops (only for test purposes). The speedup we measured is around 10%.
That gives us an upper bound on the actual runtime cost of casts.

We are currently working on producing benchmarks on real sized programs, it is not easy as Wasocaml is still a
prototype and is not yet integrated into existing build systems.

Compared to a JavaScript VM, a Wasm compiler is a much simpler beast that can compile ahead of time. For this
reason, various Wasm engines are expected to behave quite similarly. They do not show any of the wild impredictabil-
ity that browsers tend to demonstrate with JavaScript. Indeed, compiling OCaml to JS using js_of_ocaml leads to
results that are usually also twice as slow as native code in the best cases, but can sometimes be much slower in an
unpredictable fashion.

Currently there is no other Wasm runtime supporting all the extensions we require. SpiderMonkey does not have
tail calls. The reference interpreter implementation of the various extensions are split in separate repositories and
merging them requires some work.

6 Perspectives

As the first version of the implementation was intended as a demonstration, it remains a bit rough around the edges.
In particular only a fraction of the externals from the standard library externals are provided as handwritten Wasm.
The only unsupported part of the language is the object-oriented fragment, by lack of time rather than due to any
specific complexity: in Flambda this fragment is reduced to a single primitive. The source code of Wasocaml is publicly
available [1].

6.1 FFI

A lot of OCaml code in the wild interacts with C or JavaScript code through bindings written using dedicated FFI
mechanisms. We plan to allow re-use of existing bindings when compiling to Wasm. When targeting the browser
through JavaScript, it is not possible to re-use C bindings: people either have to rewrite their code in pure OCaml, or
to use JavaScript bindings. Compiling C bindings directly to Wasm allows to re-use them as-is.

C bindings. Some extensions recently added to Clang [6] allow to compile C code to Wasm in a way that makes
reusing existing OCaml bindings to C code possible with almost no change. We would only have to provide a modified
version of the OCaml native FFI headers files, replacing the usual macros with hand written Wasm functions. The only
limitation that we forsee is that the Fieldmacro won’t be an l-value anymore; a new Set_fieldmacro will be needed
instead.

JavaScript bindings. To re-use existing bindings of OCaml to JavaScript code, we need one more extension: the
reference-typed strings proposal [24]. Almost all external calls in the JavaScript FFI go through the Js.Unsafe.meth_call
function which has the type 'a -> string -> any array -> 'b. It can be exposed to the Wasm module by the em-
bedder as a function of type:

(func $meth_call

(param $obj externref)

(param $method stringref)

(param $args $anyarray)

(result externref))

This calls a method of name $method on the object $obj with the arguments $args. The JavaScript side manages
all the dynamic typing.

6.2 Effect Handlers Support

Our compiler is based on OCaml 4.14. The OCaml 5 compiler introduces effect handlers [10], a mechanism that can be
seen as a generalisation of exceptions, allowing them to be resumed. They were out of scope for our compiler, yet we
describe three strategies to handle them in the future.

CPS Compilation. It is possible to represent effect handlers as continuations, using whole-program CPS transforma-
tion [4]. However it is not ideal for two reasons. It requires the program to contain OCaml code only. Moreover, it
prevents the use of the stack of the target language and instead stores that stack in the heap, with a non-negligible cost.
It can be mitigated by only applying the transformation to code that cannot be proved to not use effect handlers but
then it is no longer compatible with separate compilation and the optimisation needs to be done as a whole program
transformation. This is the choice made by Js_of_ocaml.

Wasm Stack Switching. There is an ongoing proposal, called stack switching [23], that exactly matches the needs for
OCaml effect handlers. With it the translation would be quite straightforward.

JavaScript Promise Integration. Another strategy is available for runtimes providing both Wasm and JavaScript.
There is another ongoing proposal called JavaScript promise integration [22]. With this proposal, effects handlers
Manuscript submitted to ACM

can be implemented with a JavaScript device. This proposal is likely to land before the proper stack switching one. It
might be a temporary solution.

7 Related Work

7.1 Garbage Collected Languages to Wasm

7.1.1 Targeting Wasm1.

Go. Go compiles to Wasm1 [9]. In order to be able to re-use the Go GC, it must be able to inspect the stack. This is
not possible within Wasm. Thus it has to maintain the stack by itself using the memory, which is much slower than
using the Wasm stack.

Haskell. Haskell also compiles toWasm [11]. It has constraints similar to Go and uses similar solutions. In particular,
it also uses the memory to manage the stack.

7.1.2 Targeting Wasm-GC.

Dart. Dart [13] compiles to Wasm-GC. It does not uses i31ref and boxes scalars in a struct.

Scheme. The last addition to the small family of compiler targeting Wasm-GC is the Guile Scheme compiler. Scheme
has many similar constraints as OCaml and Guile uses many similar solutions developed independently of Wasocaml.
The compiler was presented to the Wasm-GC working group [26]. A more detailed explanation is published [25]. The
implementation and its in-depth technical description are available [27].

7.2 OCaml Web Compilers

The history of OCaml compilers targeting web languages is quite crowded. This is surely thanks to the great pleasure
that is writing compilers in OCaml.

7.2.1 Targeting JavaScript. There are multiple OCaml compilers targeting JavaScript. Many approaches where experi-
mented, the main two live ones are js_of_ocaml and melange. Naive compilation of OCaml to JavaScript is quite simple
as it can almost be summarized as “type erasure”. There are some limitations in JavaScript that prevent that to be a
complete compilation strategy. Moreover, getting to a proper compiler producing efficient and small JavaScript code
is more complex.

Jsoo. Jsoo [15] tries to be as close as possible to the native semantics. It compiles OCaml bytecode programs to
JavaScript by decompiling it back to a simple untyped lambda calculus then performs many optimisation and minimi-
sation passes.

Melange. Melange [8] tries to produce readable JavaScript, with a closer integration in the JavaScript module system.
Melange starts from a modified version of the lambda IR which provides more type information. This allows the use
of JavaScript features that match the uses of source features at the cost of some small semantical differences.

7.2.2 Targeting Wasm.

OCamlrun WebAssembly. OCamlrun WebAssembly [7] compiles the OCaml runtime and interpreter to Wasm.They
are both written in C so it is possible without anyWasm extension. In some cases, it allows to start executing code faster

than when compilin OCaml to Wasm. This is because there is less WebAssembly code to compile for the embedder.
On the other hand, the execution is slower as it is interpreting OCaml bytecode inside Wasm instead of running a
compiled version.

WASICaml. WASICaml [12] is quite similar to OCamlrun WebAssembly, the main difference being that WASICaml
does not interpret the bytecode directly but translates it partially to Wasm. It leads to a faster execution.

These approaches suffer from the same problem as C programs running on Wasm: they cannot natively manipulate
external values, such as DOM objects in the browser. Indeed, in the first versions of Wasm, the only types are scalars
(i32, i64, f32 and f64). There is no way to directly manipulate values from the embedder (e.g. JavaScript objects). It
could be possible to identify objects with an integer and map them to their corresponding objects on the embedder side.
This approach comes with the usual limitations of having two runtimes manipulating (indirectly) garbage-collected
values: it is tedious and easily leads to memory leaks (cycles between the two runtimes cannot be collected).

Wasm_of_ocaml. Wasm_of_ocaml [14] is a fork of Js_of_ocaml. It compiles the bytecode to Wasm-GC. It was de-
veloped after Wasocaml and is based on the techniques we developed and that are described in the article. It does not
benefit from the optimisations provided by Flambda. It is also quite restricted in its value representation choice and
must use a representation based on arrays.

8 Conclusion

The original goal of this implementation was to validate the Wasm-GC proposal and discuss its limits. This part was
successful. No major changes were required for OCaml (mainly keeping the i31ref type and changes to the maximal
length of sub-typing chains). And we are quite happy to claim that Wasm-GC is, to our opinion a very good design,
that is a perfect compilation target for a garbage collected functional language like OCaml. And we think that the
experience for most other garbage-collected languages will probably be similar.

As a side-effect, we now have an OCaml to Wasm-GC compiler. It is not yet usable because there are no user-side
Wasm-GC runtime available. In order to test our compiler we are using V8 with various flags to enable experimental
features such as the GC support. The design of the various extensions required by Wasocaml is stable and quite close
to completion, but some details are still in flux. For this reason, we cannot expect them to be widely available soon. On
the other hand, it means that our compiler will be ready when browsers start deploying new Wasm extensions.

Our future plans are to complete the Wasm implementations of OCaml externals, to implement the various FFI
mechanisms, support effect handlers and to move Wasocaml to Flambda2. All of these will allow easy deployment of
multi-language software on the browser while having good and predictable performances.

